Image Analysis with Fiji
Challenge Answers

Section 2

There are many possible solutions to this problem. Here is an example workflow:

- Apply a Laplacian of Gaussian filter to the image. This can be implemented with a difference of Gaussians (filter sizes of 1.5 and 2.1 respectively work well).
- Apply an automated threshold such as Otsu.
- Process the binary image by filling holes and using a watershed transformation to separate touching objects.
- Using the Analyze Particles plugin to count spots

Note that you may disagree with the some of the spot detections from the above workflow. For example you may think a spot should have been detected when is has not (false negative) or that it has been detected when it shouldn't (false positive). However if you ask your neighbour they may disagree in some cases. You might even disagree with yourself if you try it again tomorrow! Automated workflows have the key advantage of being unbiased and will return the same result each time.

Section 3

This part of this problem requires the use of a "for loop" to select each slice in a z-stack and draw some text. The following macro is a typical example:

```
// open mri-stack.tif
open("C:\\Users\\pike01\\Documents\\Reports\\FIJI Course\\sample data\\mri-stack.tif");

// get the total number of slices and store as a variable
var n = nSlices();

// loop through all slices
for (i = 1; i <= n; i = i + 1) {

		// set the slice
		setSlice(i);

		// draw some text at position (5, 5)
		makeText(i, 5, 5);
		run("Draw", "slice");
}
```

Use the command recorder to generate the macro for the second part of the problem. Then use getNumber() to take user input for the smaller sigma. Calculate the bigger sigma using the smaller and modify the Gaussian blurs to take these sigma as inputs:

```
//close all windows before starting
run("Close All");


// get smaller sigma from user input
var smallSigma = getNumber("sigma for small Gaussian:", 1.5);

// calculate bigger sigma from smaller
var bigSigma = smallSigma * 1.4;

// open helaCellsTwoChannel.tif
open("C:\\Users\\pike01\\Documents\\Reports\\FIJI Course\\sample data\\helaCellsTwoChannel.tif");

//split channels into separate windows
run("Split Channels");

//select and close the window for the second channel
selectWindow("C2-helaCellsTwoChannel.tif");
close();

//select 1st channel
selectWindow("C1-helaCellsTwoChannel.tif");

// duplicate channel two twice and rename new windows as smallGaussian and bigGuassian
run("Duplicate...", "title=smallGaussian");
selectWindow("C1-helaCellsTwoChannel.tif");
run("Duplicate...", "title=bigGaussian");

// select smallGuassian and blur with Gaussian (sigma=smallSigma)
selectWindow("smallGaussian");
run("Gaussian Blur...", "sigma=" + smallSigma);

// select bigGuassian and blur with Gaussian (sigma=bigSigma)
selectWindow("bigGaussian");
run("Gaussian Blur...", "sigma=" + bigSigma);

// subract smallGuassian from bigGuassian
imageCalculator("Subtract create", "smallGaussian","bigGaussian");

// select filtered image and Otsu threshold
selectWindow("Result of smallGaussian");
setAutoThreshold("Otsu dark");
setOption("BlackBackground", false) ;
run("Convert to Mask");

// Watershed Transformation to Seperate touching objects
run("Watershed");

// Analyze Particles plugin with minimum size of 3 pixels
run("Analyze Particles...", "size=3-Infinity pixel exclude clear summarize add");
```